// File: contracts/lib/openzeppelin-contracts/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: contracts/lib/openzeppelin-contracts/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: contracts/lib/openzeppelin-contracts/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: contracts/lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: contracts/lib/openzeppelin-contracts/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/lib/openzeppelin-contracts/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: contracts/lib/openzeppelin-contracts/contracts/utils/cryptography/ECDSA.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/cryptography/ECDSA.sol)

pragma solidity ^0.8.0;

import "../Strings.sol";

/**
 * @dev Elliptic Curve Digital Signature Algorithm (ECDSA) operations.
 *
 * These functions can be used to verify that a message was signed by the holder
 * of the private keys of a given address.
 */
library ECDSA {
    enum RecoverError {
        NoError,
        InvalidSignature,
        InvalidSignatureLength,
        InvalidSignatureS,
        InvalidSignatureV // Deprecated in v4.8
    }

    function _throwError(RecoverError error) private pure {
        if (error == RecoverError.NoError) {
            return; // no error: do nothing
        } else if (error == RecoverError.InvalidSignature) {
            revert("ECDSA: invalid signature");
        } else if (error == RecoverError.InvalidSignatureLength) {
            revert("ECDSA: invalid signature length");
        } else if (error == RecoverError.InvalidSignatureS) {
            revert("ECDSA: invalid signature 's' value");
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature` or error string. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     *
     * Documentation for signature generation:
     * - with https://web3js.readthedocs.io/en/v1.3.4/web3-eth-accounts.html#sign[Web3.js]
     * - with https://docs.ethers.io/v5/api/signer/#Signer-signMessage[ethers]
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, bytes memory signature) internal pure returns (address, RecoverError) {
        if (signature.length == 65) {
            bytes32 r;
            bytes32 s;
            uint8 v;
            // ecrecover takes the signature parameters, and the only way to get them
            // currently is to use assembly.
            /// @solidity memory-safe-assembly
            assembly {
                r := mload(add(signature, 0x20))
                s := mload(add(signature, 0x40))
                v := byte(0, mload(add(signature, 0x60)))
            }
            return tryRecover(hash, v, r, s);
        } else {
            return (address(0), RecoverError.InvalidSignatureLength);
        }
    }

    /**
     * @dev Returns the address that signed a hashed message (`hash`) with
     * `signature`. This address can then be used for verification purposes.
     *
     * The `ecrecover` EVM opcode allows for malleable (non-unique) signatures:
     * this function rejects them by requiring the `s` value to be in the lower
     * half order, and the `v` value to be either 27 or 28.
     *
     * IMPORTANT: `hash` _must_ be the result of a hash operation for the
     * verification to be secure: it is possible to craft signatures that
     * recover to arbitrary addresses for non-hashed data. A safe way to ensure
     * this is by receiving a hash of the original message (which may otherwise
     * be too long), and then calling {toEthSignedMessageHash} on it.
     */
    function recover(bytes32 hash, bytes memory signature) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, signature);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `r` and `vs` short-signature fields separately.
     *
     * See https://eips.ethereum.org/EIPS/eip-2098[EIP-2098 short signatures]
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, bytes32 r, bytes32 vs) internal pure returns (address, RecoverError) {
        bytes32 s = vs & bytes32(0x7fffffffffffffffffffffffffffffffffffffffffffffffffffffffffffffff);
        uint8 v = uint8((uint256(vs) >> 255) + 27);
        return tryRecover(hash, v, r, s);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `r and `vs` short-signature fields separately.
     *
     * _Available since v4.2._
     */
    function recover(bytes32 hash, bytes32 r, bytes32 vs) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, r, vs);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Overload of {ECDSA-tryRecover} that receives the `v`,
     * `r` and `s` signature fields separately.
     *
     * _Available since v4.3._
     */
    function tryRecover(bytes32 hash, uint8 v, bytes32 r, bytes32 s) internal pure returns (address, RecoverError) {
        // EIP-2 still allows signature malleability for ecrecover(). Remove this possibility and make the signature
        // unique. Appendix F in the Ethereum Yellow paper (https://ethereum.github.io/yellowpaper/paper.pdf), defines
        // the valid range for s in (301): 0 < s < secp256k1n ÷ 2 + 1, and for v in (302): v ∈ {27, 28}. Most
        // signatures from current libraries generate a unique signature with an s-value in the lower half order.
        //
        // If your library generates malleable signatures, such as s-values in the upper range, calculate a new s-value
        // with 0xFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFEBAAEDCE6AF48A03BBFD25E8CD0364141 - s1 and flip v from 27 to 28 or
        // vice versa. If your library also generates signatures with 0/1 for v instead 27/28, add 27 to v to accept
        // these malleable signatures as well.
        if (uint256(s) > 0x7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF5D576E7357A4501DDFE92F46681B20A0) {
            return (address(0), RecoverError.InvalidSignatureS);
        }

        // If the signature is valid (and not malleable), return the signer address
        address signer = ecrecover(hash, v, r, s);
        if (signer == address(0)) {
            return (address(0), RecoverError.InvalidSignature);
        }

        return (signer, RecoverError.NoError);
    }

    /**
     * @dev Overload of {ECDSA-recover} that receives the `v`,
     * `r` and `s` signature fields separately.
     */
    function recover(bytes32 hash, uint8 v, bytes32 r, bytes32 s) internal pure returns (address) {
        (address recovered, RecoverError error) = tryRecover(hash, v, r, s);
        _throwError(error);
        return recovered;
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from a `hash`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes32 hash) internal pure returns (bytes32) {
        // 32 is the length in bytes of hash,
        // enforced by the type signature above
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n32", hash));
    }

    /**
     * @dev Returns an Ethereum Signed Message, created from `s`. This
     * produces hash corresponding to the one signed with the
     * https://eth.wiki/json-rpc/API#eth_sign[`eth_sign`]
     * JSON-RPC method as part of EIP-191.
     *
     * See {recover}.
     */
    function toEthSignedMessageHash(bytes memory s) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19Ethereum Signed Message:\n", Strings.toString(s.length), s));
    }

    /**
     * @dev Returns an Ethereum Signed Typed Data, created from a
     * `domainSeparator` and a `structHash`. This produces hash corresponding
     * to the one signed with the
     * https://eips.ethereum.org/EIPS/eip-712[`eth_signTypedData`]
     * JSON-RPC method as part of EIP-712.
     *
     * See {recover}.
     */
    function toTypedDataHash(bytes32 domainSeparator, bytes32 structHash) internal pure returns (bytes32) {
        return keccak256(abi.encodePacked("\x19\x01", domainSeparator, structHash));
    }
}

// File: contracts/lib/openzeppelin-contracts/contracts/utils/cryptography/EIP712.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/cryptography/EIP712.sol)

pragma solidity ^0.8.0;

import "./ECDSA.sol";

/**
 * @dev https://eips.ethereum.org/EIPS/eip-712[EIP 712] is a standard for hashing and signing of typed structured data.
 *
 * The encoding specified in the EIP is very generic, and such a generic implementation in Solidity is not feasible,
 * thus this contract does not implement the encoding itself. Protocols need to implement the type-specific encoding
 * they need in their contracts using a combination of `abi.encode` and `keccak256`.
 *
 * This contract implements the EIP 712 domain separator ({_domainSeparatorV4}) that is used as part of the encoding
 * scheme, and the final step of the encoding to obtain the message digest that is then signed via ECDSA
 * ({_hashTypedDataV4}).
 *
 * The implementation of the domain separator was designed to be as efficient as possible while still properly updating
 * the chain id to protect against replay attacks on an eventual fork of the chain.
 *
 * NOTE: This contract implements the version of the encoding known as "v4", as implemented by the JSON RPC method
 * https://docs.metamask.io/guide/signing-data.html[`eth_signTypedDataV4` in MetaMask].
 *
 * _Available since v3.4._
 */
abstract contract EIP712 {
    /* solhint-disable var-name-mixedcase */
    // Cache the domain separator as an immutable value, but also store the chain id that it corresponds to, in order to
    // invalidate the cached domain separator if the chain id changes.
    bytes32 private immutable _CACHED_DOMAIN_SEPARATOR;
    uint256 private immutable _CACHED_CHAIN_ID;
    address private immutable _CACHED_THIS;

    bytes32 private immutable _HASHED_NAME;
    bytes32 private immutable _HASHED_VERSION;
    bytes32 private immutable _TYPE_HASH;

    /* solhint-enable var-name-mixedcase */

    /**
     * @dev Initializes the domain separator and parameter caches.
     *
     * The meaning of `name` and `version` is specified in
     * https://eips.ethereum.org/EIPS/eip-712#definition-of-domainseparator[EIP 712]:
     *
     * - `name`: the user readable name of the signing domain, i.e. the name of the DApp or the protocol.
     * - `version`: the current major version of the signing domain.
     *
     * NOTE: These parameters cannot be changed except through a xref:learn::upgrading-smart-contracts.adoc[smart
     * contract upgrade].
     */
    constructor(string memory name, string memory version) {
        bytes32 hashedName = keccak256(bytes(name));
        bytes32 hashedVersion = keccak256(bytes(version));
        bytes32 typeHash = keccak256(
            "EIP712Domain(string name,string version,uint256 chainId,address verifyingContract)"
        );
        _HASHED_NAME = hashedName;
        _HASHED_VERSION = hashedVersion;
        _CACHED_CHAIN_ID = block.chainid;
        _CACHED_DOMAIN_SEPARATOR = _buildDomainSeparator(typeHash, hashedName, hashedVersion);
        _CACHED_THIS = address(this);
        _TYPE_HASH = typeHash;
    }

    /**
     * @dev Returns the domain separator for the current chain.
     */
    function _domainSeparatorV4() internal view returns (bytes32) {
        if (address(this) == _CACHED_THIS && block.chainid == _CACHED_CHAIN_ID) {
            return _CACHED_DOMAIN_SEPARATOR;
        } else {
            return _buildDomainSeparator(_TYPE_HASH, _HASHED_NAME, _HASHED_VERSION);
        }
    }

    function _buildDomainSeparator(
        bytes32 typeHash,
        bytes32 nameHash,
        bytes32 versionHash
    ) private view returns (bytes32) {
        return keccak256(abi.encode(typeHash, nameHash, versionHash, block.chainid, address(this)));
    }

    /**
     * @dev Given an already https://eips.ethereum.org/EIPS/eip-712#definition-of-hashstruct[hashed struct], this
     * function returns the hash of the fully encoded EIP712 message for this domain.
     *
     * This hash can be used together with {ECDSA-recover} to obtain the signer of a message. For example:
     *
     * ```solidity
     * bytes32 digest = _hashTypedDataV4(keccak256(abi.encode(
     *     keccak256("Mail(address to,string contents)"),
     *     mailTo,
     *     keccak256(bytes(mailContents))
     * )));
     * address signer = ECDSA.recover(digest, signature);
     * ```
     */
    function _hashTypedDataV4(bytes32 structHash) internal view virtual returns (bytes32) {
        return ECDSA.toTypedDataHash(_domainSeparatorV4(), structHash);
    }
}

// File: contracts/lib/openzeppelin-contracts/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1, "Math: mulDiv overflow");

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(uint256 x, uint256 y, uint256 denominator, Rounding rounding) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10 ** 64) {
                value /= 10 ** 64;
                result += 64;
            }
            if (value >= 10 ** 32) {
                value /= 10 ** 32;
                result += 32;
            }
            if (value >= 10 ** 16) {
                value /= 10 ** 16;
                result += 16;
            }
            if (value >= 10 ** 8) {
                value /= 10 ** 8;
                result += 8;
            }
            if (value >= 10 ** 4) {
                value /= 10 ** 4;
                result += 4;
            }
            if (value >= 10 ** 2) {
                value /= 10 ** 2;
                result += 2;
            }
            if (value >= 10 ** 1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10 ** result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result << 3) < value ? 1 : 0);
        }
    }
}

// File: contracts/src/IUSDC.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.17;

import "openzeppelin-contracts/token/ERC20/IERC20.sol";

interface IUSDC is IERC20 {
    function receiveWithAuthorization(
        address from,
        address to,
        uint256 value,
        uint256 validAfter,
        uint256 validBefore,
        bytes32 nonce,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;
}

// File: contracts/src/IndieBrokerV1.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.17;

import "openzeppelin-contracts/utils/cryptography/ECDSA.sol";
import "openzeppelin-contracts/utils/cryptography/EIP712.sol";
import "openzeppelin-contracts/access/Ownable.sol";
import "openzeppelin-contracts/security/Pausable.sol";
import "openzeppelin-contracts/security/ReentrancyGuard.sol";
import "./IUSDC.sol";

/**
 * @title IndieBrokerV1
 * @author IndieDAO
 * @notice The Indie Protocol contract
 */
contract IndieBrokerV1 is EIP712, Ownable, Pausable, ReentrancyGuard {
    struct Project {
        uint256 projectId;
        address leadAddress;
        address clientAddress;
        address salesAddress;
    }

    struct SprintPayments {
        uint256 totalAmount;
        uint256 totalTreasuryAmount;
        uint256 totalCashVestingAmount;
        uint256 totalLeadAmount;
        uint256 totalSalesAmount;
        uint256[] payeeAmounts;
        uint256[] treasuryAmounts;
        uint256[] leadAmounts;
        uint256[] salesAmounts;
        uint256[] cashVestingAmounts;
    }

    error ExceededMaxFees();
    error ZeroAddress();
    error InvalidFee();
    error FeeTooLow();
    error FeeTooHigh();
    error NotAuthorized();
    error PayeeAmountMismatch();
    error SprintAlreadyCompleted();
    error InvalidClientSignature();
    error ProjectAlreadyExists();
    error ProjectNotFound();
    error InvalidProjectId();
    error InvalidSprintId();
    error LeadNotOnAllowList();
    error InsufficientBalance();

    event StartProject(
        uint256 indexed projectId, address indexed leadAddress, address indexed clientAddress, address salesAddress
    );

    event CompleteProjectSprint(uint256 indexed projectId, uint256 indexed sprintId, uint256 totalAmount);

    event SendDeposit(uint256 indexed projectId, address indexed sender, uint256 amount);

    event DistributePayment(
        address indexed payee,
        uint256 indexed projectId,
        uint256 totalAmount,
        uint256 payeeAmount,
        uint256 treasuryAmount,
        uint256 leadAmount,
        uint256 salesAmount,
        uint256 cashVestingAmount
    );

    event SetFee(Fees indexed f, uint256 fee);

    event SetFeeRecipient(Fees indexed f, address recipient);

    event ReassignProjectLead(uint256 indexed projectId, address newLeadAddress);

    event ReassignProjectSales(uint256 indexed projectId, address newSalesAddress);

    event ReassignProjectClient(uint256 indexed projectId, address newClientAddress);

    event SetIndividualTreasuryFee(address indexed addr, uint256 fee);

    event SetMinMaxIndividualTreasuryFees(uint256 minFee, uint256 maxFee);

    event SetAllowedLead(address indexed addr, bool allowed);

    event WithdrawFromProject(uint256 indexed projectId, uint256 amount, address indexed recipient);

    enum Fees {
        Treasury,
        Sales,
        Lead,
        CashVesting
    }

    /**
     * @notice Mapping of project IDs to project structs
     */
    mapping(uint256 => Project) public projects;

    /**
     * @notice Mapping of project IDs to project balances
     * @dev projectId => USDC balance
     */
    mapping(uint256 => uint256) public projectBalances;

    /**
     * @notice Mapping of project/sprint tuples to project sprint completion status
     * @dev projectId => sprintId => completed
     */
    mapping(uint256 => mapping(uint256 => bool)) public completedSprints;

    /**
     * @notice Mapping of fee types to fee percentages
     * @dev Fees => fee percentage
     */
    mapping(Fees => uint256) public fees;

    /**
     * @notice Mapping of fee types to fee recipients
     * @dev Fees => fee recipient address
     */
    mapping(Fees => address) public feeRecipients;

    /**
     * @notice Mapping of individual treasury fees
     * @dev address => fee percentage
     */
    mapping(address => uint256) public individualTreasuryFees;

    /**
     * @notice Mapping of allowed leads
     * @dev address => allowed
     */
    mapping(address => bool) public allowedLeads;

    /**
     * @notice Reference to the USDC contract
     */
    IUSDC public usdc;

    /**
     * @notice The minimum possible individual treasury fee
     */
    uint256 public minIndividualTreasuryFee = 20_00; // 20%

    /**
     * @notice The maximum possible individual treasury fee
     */
    uint256 public maxIndividualTreasuryFee = 20_00; // 20%

    uint256 internal constant MAX_FEES = 100_00; // The sum of fees cannot exceed 100%
    uint256 internal constant FEE_DENOMINATOR = 100_00;

    /**
     * @notice Constructor
     * @param usdc_ The USDC contract address
     * @param treasuryAddress_ The treasury address
     * @param cashVestingAddress_ The cash vesting address
     */
    constructor(IUSDC usdc_, address treasuryAddress_, address cashVestingAddress_) EIP712("IndieBroker", "1") {
        usdc = usdc_;
        feeRecipients[Fees.Treasury] = treasuryAddress_;
        feeRecipients[Fees.CashVesting] = cashVestingAddress_;

        fees[Fees.Treasury] = 20_00; // 20%
        fees[Fees.Lead] = 10_00; // 10%
        fees[Fees.Sales] = 10_00; // 10%
        fees[Fees.CashVesting] = 5_00; // 5%
    }

    /**
     * @notice Starts a project
     * @param projectId The reference project identifier from the off-chain Indie Protocol database
     * @param leadAddress The project lead address
     * @param clientAddress The project client address
     * @param salesAddress The project sales referrer address
     */
    function startProject(uint256 projectId, address leadAddress, address clientAddress, address salesAddress)
        external
        whenNotPaused
        nonReentrant
    {
        if (projectId == 0) {
            revert InvalidProjectId();
        }

        if (leadAddress == address(0)) {
            revert ZeroAddress();
        }

        if (clientAddress == address(0)) {
            revert ZeroAddress();
        }

        if (salesAddress == address(0)) {
            revert ZeroAddress();
        }

        if (msg.sender != leadAddress) {
            revert NotAuthorized();
        }

        if (!allowedLeads[leadAddress]) {
            revert LeadNotOnAllowList();
        }

        if (projects[projectId].projectId != 0) {
            revert ProjectAlreadyExists();
        }

        projects[projectId] = Project(projectId, leadAddress, clientAddress, salesAddress);
        emit StartProject(projectId, leadAddress, clientAddress, salesAddress);
    }

    /**
     * @notice Sends a deposit to the Indie Protocol contract for a specific project
     * @dev See USDC contract `receiveWithAuthorization` function for more details
     * @param projectId The reference project identifier from the off-chain Indie Protocol database
     * @param from The address to transfer the USDC from
     * @param amount The amount of USDC to transfer
     * @param validAfter The time after which this is valid (unix time)
     * @param validBefore The time before which this is valid (unix time)
     * @param nonce A unique nonce
     * @param v The recovery byte of the signature
     * @param r Half of the ECDSA signature pair
     * @param s Half of the ECDSA signature pair
     */
    function sendDepositWithAuthorization(
        uint256 projectId,
        address from,
        uint256 amount,
        uint256 validAfter,
        uint256 validBefore,
        bytes32 nonce,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external whenNotPaused nonReentrant {
        _sendDeposit(projectId, from, amount);

        usdc.receiveWithAuthorization(from, address(this), amount, validAfter, validBefore, nonce, v, r, s);
    }

    /**
     * @notice Sends a deposit to the Indie Protocol contract for a specific project
     * @dev See USDC contract `transferFrom` function for more details
     * @param projectId The reference project identifier from the off-chain Indie Protocol database
     * @param from The address to transfer the USDC from
     * @param amount The amount of USDC to transfer
     */
    function sendDeposit(uint256 projectId, address from, uint256 amount) external whenNotPaused nonReentrant {
        _sendDeposit(projectId, from, amount);

        usdc.transferFrom(from, address(this), amount);
    }

    /**
     * @notice Calculates payment amounts given a set of payees and amounts
     * @param payees The addresses of the payees
     * @param amounts The amounts of USDC expected to be distributed for each payee
     * @return sprintPayments The calculated sprint payments (see `SprintPayments` struct)
     */
    function calculateSprintPayments(address[] calldata payees, uint256[] calldata amounts)
        public
        view
        returns (SprintPayments memory sprintPayments)
    {
        sprintPayments = SprintPayments(
            0, // totalAmount
            0, // totalTreasuryAmount
            0, // totalCashVestingAmount
            0, // totalLeadAmount
            0, // totalSalesAmount
            new uint256[](payees.length), // payeeAmounts
            new uint256[](payees.length), // treasuryAmounts
            new uint256[](payees.length), // leadAmounts
            new uint256[](payees.length), // salesAmounts
            new uint256[](payees.length) // cashVestingAmounts
        );

        uint256 payeeAmount;
        uint256 treasuryAmount;
        uint256 leadAmount;
        uint256 salesAmount;
        uint256 cashVestingAmount;

        for (uint256 i = 0; i < payees.length;) {
            sprintPayments.totalAmount += amounts[i];

            (payeeAmount, treasuryAmount, leadAmount, salesAmount, cashVestingAmount) =
                calculatePayment(payees[i], amounts[i]);

            sprintPayments.totalTreasuryAmount += treasuryAmount;
            sprintPayments.totalLeadAmount += leadAmount;
            sprintPayments.totalSalesAmount += salesAmount;
            sprintPayments.totalCashVestingAmount += cashVestingAmount;

            sprintPayments.payeeAmounts[i] = payeeAmount;
            sprintPayments.treasuryAmounts[i] = treasuryAmount;
            sprintPayments.leadAmounts[i] = leadAmount;
            sprintPayments.salesAmounts[i] = salesAmount;
            sprintPayments.cashVestingAmounts[i] = cashVestingAmount;

            unchecked {
                i++;
            }
        }
    }

    /**
     * @notice Calculates payment distributions given a payee and amount
     * @param payee The address of the payee
     * @param amount The amount of USDC expected to be distributed
     * @return payeeAmount The amount of USDC expected to be distributed to the payee after rewards are deducted
     * @return treasuryAmount The amount of USDC expected to be distributed to the treasury
     * @return leadAmount The amount of USDC expected to be distributed to the project lead (only includes the lead reward)
     * @return salesAmount The amount of USDC expected to be distributed to the project sales referrer
     * @return cashVestingAmount The amount of USDC expected to be distributed to the cash vesting address
     */
    function calculatePayment(address payee, uint256 amount)
        public
        view
        returns (
            uint256 payeeAmount,
            uint256 treasuryAmount,
            uint256 leadAmount,
            uint256 salesAmount,
            uint256 cashVestingAmount
        )
    {
        uint256 treasuryFee = _getTreasuryFeeForIndividual(payee);
        treasuryAmount = (amount * treasuryFee) / FEE_DENOMINATOR;
        leadAmount = (amount * fees[Fees.Lead]) / FEE_DENOMINATOR;
        salesAmount = (amount * fees[Fees.Sales]) / FEE_DENOMINATOR;
        cashVestingAmount = (amount * fees[Fees.CashVesting]) / FEE_DENOMINATOR;
        payeeAmount = amount - treasuryAmount - leadAmount - salesAmount - cashVestingAmount;
    }

    /**
     * @notice Completes a sprint and distributes the funds to the payees and reward recipients
     * @param projectId The reference project identifier from the off-chain Indie Protocol database
     * @param sprintId The reference sprint identifier from the off-chain Indie Protocol database
     * @param payees The addresses of the payees
     * @param amounts The amounts of USDC expected to be distributed for each payee
     * @param v The recovery byte of the client's acceptance signature
     * @param r Half of the ECDSA signature pair
     * @param s Half of the ECDSA signature pair
     */
    function completeSprint(
        uint256 projectId,
        uint256 sprintId,
        address[] calldata payees,
        uint256[] calldata amounts,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external whenNotPaused nonReentrant {
        if (sprintId == 0) {
            revert InvalidSprintId();
        }

        if (payees.length != amounts.length) {
            revert PayeeAmountMismatch();
        }

        if (completedSprints[projectId][sprintId]) {
            revert SprintAlreadyCompleted();
        }

        Project memory project = projects[projectId];
        if (msg.sender != project.leadAddress) {
            revert NotAuthorized();
        }

        SprintPayments memory sprintPayments = calculateSprintPayments(payees, amounts);

        bytes32 structHash = keccak256(
            abi.encode(
                keccak256("CompleteSprint(uint256 projectId,uint256 sprintId,uint256 totalAmount)"),
                projectId,
                sprintId,
                sprintPayments.totalAmount
            )
        );

        bytes32 digest = _hashTypedDataV4(structHash);

        if (!_verifySignature(digest, v, r, s, project.clientAddress)) {
            revert InvalidClientSignature();
        }

        for (uint256 i = 0; i < sprintPayments.payeeAmounts.length;) {
            if (payees[i] == address(0)) {
                revert ZeroAddress();
            }

            emit DistributePayment(
                payees[i],
                projectId,
                amounts[i],
                sprintPayments.payeeAmounts[i],
                sprintPayments.treasuryAmounts[i],
                sprintPayments.leadAmounts[i],
                sprintPayments.salesAmounts[i],
                sprintPayments.cashVestingAmounts[i]
                );

            if (sprintPayments.payeeAmounts[i] > 0) {
                usdc.transfer(payees[i], sprintPayments.payeeAmounts[i]);
            }

            unchecked {
                i++;
            }
        }

        if (sprintPayments.totalTreasuryAmount > 0) {
            usdc.transfer(feeRecipients[Fees.Treasury], sprintPayments.totalTreasuryAmount);
        }

        if (sprintPayments.totalCashVestingAmount > 0) {
            usdc.transfer(feeRecipients[Fees.CashVesting], sprintPayments.totalCashVestingAmount);
        }

        if (sprintPayments.totalLeadAmount > 0) {
            usdc.transfer(project.leadAddress, sprintPayments.totalLeadAmount);
        }

        if (sprintPayments.totalSalesAmount > 0) {
            usdc.transfer(project.salesAddress, sprintPayments.totalSalesAmount);
        }

        projectBalances[projectId] -= sprintPayments.totalAmount;
        completedSprints[projectId][sprintId] = true;

        emit CompleteProjectSprint(projectId, sprintId, sprintPayments.totalAmount);
    }

    /**
     * @notice Sets the fee for an individual payee address
     * @param addr The address of the payee
     * @param fee The fee to be set where the value has a denominator of 10000 (e.g. `500` yields a 5% fee). See `minIndividualTreasuryFee` and `maxIndividualTreasuryFee` for limits.
     */
    function setIndividualTreasuryFee(address addr, uint256 fee) external {
        if (addr != msg.sender) {
            revert NotAuthorized();
        }
        if (fee < minIndividualTreasuryFee) {
            revert FeeTooLow();
        }
        if (fee > maxIndividualTreasuryFee) {
            revert FeeTooHigh();
        }

        individualTreasuryFees[addr] = fee;

        emit SetIndividualTreasuryFee(addr, fee);
    }

    /**
     * @notice Sets the fee for a given fee type
     * @param f The fee type to be set
     * @param fee The fee to be set where the value has a denominator of 10000 (e.g. `500` yields a 5% fee)
     */
    function setFee(Fees f, uint256 fee) external onlyOwner {
        fees[f] = fee;

        // Ensure the sum of all possible fees does not exceed the max possible
        // Must account for the possibility of an individual maxing out their invididual treasury fee
        uint256 maxTreasuryFee = Math.max(fees[Fees.Treasury], maxIndividualTreasuryFee);

        if (maxTreasuryFee + fees[Fees.Lead] + fees[Fees.Sales] + fees[Fees.CashVesting] > MAX_FEES) {
            revert ExceededMaxFees();
        }

        emit SetFee(f, fee);
    }

    /**
     * @notice Sets the fee recipient for a given fee type
     * @param f The fee type to be set must be Treasury or CashVesting
     * @param recipient The address of the recipient
     */
    function setFeeRecipient(Fees f, address recipient) external onlyOwner {
        if (recipient == address(0)) {
            revert ZeroAddress();
        }
        if (f != Fees.Treasury && f != Fees.CashVesting) {
            revert InvalidFee();
        }

        feeRecipients[f] = recipient;
        emit SetFeeRecipient(f, recipient);
    }

    /**
     * @notice Sets the minimum and maximum individual treasury fees
     * @param minFee The minimum fee to be set where the value has a denominator of 10000 (e.g. `500` yields a 5% fee)
     * @param maxFee The maximum fee to be set where the value has a denominator of 10000 (e.g. `500` yields a 5% fee)
     */
    function setMinMaxIndividualTreasuryFees(uint256 minFee, uint256 maxFee) external onlyOwner {
        if (minFee < 0) {
            revert FeeTooLow();
        }
        if (minFee > maxFee) {
            revert FeeTooHigh();
        }

        // Max individual fee plus all other (non-treasury) fees cannot exceed max fees
        if (maxFee + fees[Fees.Lead] + fees[Fees.Sales] + fees[Fees.CashVesting] > MAX_FEES) {
            revert ExceededMaxFees();
        }

        minIndividualTreasuryFee = minFee;
        maxIndividualTreasuryFee = maxFee;

        emit SetMinMaxIndividualTreasuryFees(minFee, maxFee);
    }

    /**
     * @notice Reassigns the project lead for a project. Can only be called by the current project lead or the contract owner.
     * @param projectId The reference project identifier from the off-chain Indie Protocol database
     * @param newLeadAddress The address of the new project lead
     */
    function reassignProjectLead(uint256 projectId, address newLeadAddress) external {
        if (newLeadAddress == address(0)) {
            revert ZeroAddress();
        }

        if (!allowedLeads[newLeadAddress]) {
            revert LeadNotOnAllowList();
        }

        Project memory project = projects[projectId];

        if (project.projectId == 0) {
            revert ProjectNotFound();
        }

        if (msg.sender != project.leadAddress && msg.sender != owner()) {
            revert NotAuthorized();
        }

        projects[projectId].leadAddress = newLeadAddress;
        emit ReassignProjectLead(projectId, newLeadAddress);
    }

    /**
     * @notice Reassigns the project sales referrer for a project. Can only be called by the current project sales referrer or the contract owner.
     * @param projectId The reference project identifier from the off-chain Indie Protocol database
     * @param newSalesAddress The address of the new project sales referrer
     */
    function reassignProjectSales(uint256 projectId, address newSalesAddress) external {
        if (newSalesAddress == address(0)) {
            revert ZeroAddress();
        }

        Project memory project = projects[projectId];

        if (project.projectId == 0) {
            revert ProjectNotFound();
        }

        if (msg.sender != project.salesAddress && msg.sender != owner()) {
            revert NotAuthorized();
        }

        projects[projectId].salesAddress = newSalesAddress;
        emit ReassignProjectSales(projectId, newSalesAddress);
    }

    /**
     * @notice Reassigns the project client for a project. Can only be called by the current project lead, client or the contract owner.
     * @param projectId The reference project identifier from the off-chain Indie Protocol database
     * @param newClientAddress The address of the new project client
     */
    function reassignProjectClient(uint256 projectId, address newClientAddress) external {
        if (newClientAddress == address(0)) {
            revert ZeroAddress();
        }

        Project memory project = projects[projectId];

        if (project.projectId == 0) {
            revert ProjectNotFound();
        }

        if (msg.sender != project.clientAddress && msg.sender != project.leadAddress && msg.sender != owner()) {
            revert NotAuthorized();
        }

        projects[projectId].clientAddress = newClientAddress;
        emit ReassignProjectClient(projectId, newClientAddress);
    }

    /**
     * @notice Enables/disables an address from being a project lead and being able to create projects
     * @param addr The address of the individual
     * @param allowed Whether or not the address is allowed to be a project lead
     */
    function setAllowedLead(address addr, bool allowed) external onlyOwner {
        if (addr == address(0)) {
            revert ZeroAddress();
        }

        allowedLeads[addr] = allowed;
        emit SetAllowedLead(addr, allowed);
    }

    /**
     * @notice Withdraws funds from a project's balance
     * @param projectId The reference project identifier from the off-chain Indie Protocol database
     * @param amount The amount to withdraw
     * @param recipient The address to send the funds to
     */
    function withdrawFromProject(uint256 projectId, uint256 amount, address recipient)
        external
        onlyOwner
        nonReentrant
    {
        if (recipient == address(0)) {
            revert ZeroAddress();
        }

        if (projectBalances[projectId] < amount) {
            revert InsufficientBalance();
        }

        projectBalances[projectId] -= amount;
        usdc.transfer(recipient, amount);

        emit WithdrawFromProject(projectId, amount, recipient);
    }

    /**
     * @notice Pauses the contract
     */
    function pause() external onlyOwner {
        _pause();
    }

    /**
     * @notice Unpauses the contract
     */
    function unpause() external onlyOwner {
        _unpause();
    }

    function _verifySignature(bytes32 digest, uint8 v, bytes32 r, bytes32 s, address signerAddress)
        internal
        pure
        returns (bool)
    {
        (address signedHashAddress, ECDSA.RecoverError error) = ECDSA.tryRecover(digest, v, r, s);

        return error == ECDSA.RecoverError.NoError && signedHashAddress == signerAddress;
    }

    function _getTreasuryFeeForIndividual(address addr) internal view returns (uint256) {
        return individualTreasuryFees[addr] == 0 ? fees[Fees.Treasury] : individualTreasuryFees[addr];
    }

    function _sendDeposit(uint256 projectId, address from, uint256 amount) internal {
        if (projects[projectId].projectId == 0) {
            revert ProjectNotFound();
        }
        projectBalances[projectId] += amount;

        emit SendDeposit(projectId, from, amount);
    }
}